
Convex Quadratic Programming in AMPL

A surprising variety of optimization 
applications can be written in terms 
of convex quadratic objectives and 
constraints that are handled 
effectively by extensions to linear 
solvers. “Elliptical” convex quadratic 
programs are easily recognized once 
the matrices of quadratic coefficients 
are extracted, through a test for 
positive-semidefiniteness. “Conic” 
problems are also convex quadratic 
and can in principle also be detected 
numerically, but are more commonly 
recognized by their equivalence to 
certain canonical forms. Additionally, 
varied combinations of sums-of-
squares, Euclidean norms, quadratic-
linear ratios, products of powers, 
p-norms, and log-Chebychev terms 

can be identified symbolically and 
transformed to quadratic problems 
that have conic formulations. The 
power and convenience of an 
algebraic modeling language may be 
extended to support these cases, with 
the help of a recursive tree-walk 
approach that detects and (where 
necessary) transforms arbitrarily 
complex instances; modelers are 
thereby freed from the time-
consuming and error-prone work of 
maintaining the equivalent canonical 
formulations explicitly. We describe 
the challenges of creating the requisite 
detection and transformation routines 
for the AMPL language, and report 
computational tests that suggest the 
usefulness of these routines.

What is convex quadratic?
 What kind of solver do you want to use?

Introductory examples
 Product of linear terms

 Traffic network

Detection and transformation
 Where they are done now

 Where they should be done

 Our new extensions
 Theory
 Implementation
 Testing

Outline

Convex quadratic objective
 PSD quadratic + linear

Convex quadratic constraints
 Linear

 PSD quadratic  ≤  constant
 Conic quadratic

Anything transformable to the above

What kind of solver do you want to use?

4

What is Convex Quadratic?

MINOS, KNITRO, Ipopt, SNOPT, CONOPT, . . .

Advantages
 Accepts any form of problem

 Tolerates nonconvexities

Disadvantages
 Relies on smoothness

 Uses complex mechanisms
 Function evals, line searches, convergence tests, . . .

 Reports only local optimality

General Nonlinear Solver
What kind of solver?

CPLEX, Gurobi, Xpress, MOSEK, . . .

Advantages
 Uses mechanisms adapted from linear programming

 Sparse coefficient lists, fast interior-point methods, . . .

 Tolerates nonsmooth functions & regions

 Reports global optimality

Disadvantages
 Requires recognizable convex quadratic formulations

 Rejects problems not in required form

Extended Linear Solver
What kind of solver?

Zero-one
 Extend linear branch-and-bound

 Transform to linear
 requires just one binary in each quadratic term
 many alternatives available

 Transform to PSD quadratic
 based on ݔଶ = ݔ for any binary ݔ

General integer
 Extend linear branch-and-bound

 Transform to zero-one
 creates logଶ ܷ binaries for domain of size ܷ

8

Possibilities for Integer Variables

Example 1: Product of Linear Terms
Original formulation

 Maximize  (∑ ௝ܿݔ௝) ௡௝ୀଵ (∑ ௝݀ݕ௝) ௡௝ୀଵ
 ∑ ௝ܿݔ௝ ≥ 0,௡௝ୀଵ   ∑ ௝݀ݕ௝ ≥ 0௡௝ୀଵ

Conic reformulation
 Maximize  ݖ
 ଶݖ ≤ ௫ݖ  ,௬ݖ ௫ݖ ≥ ௬ݖ ,0 ≥ 0
 ௫ݖ = ∑ ௝ܿݔ௝, ௡௝ୀଵ ௬ݖ  = ∑ ௝݀ݕ௝ ௡௝ୀଵ

AMPL Model
Direct formulation

param n > 0;
param c {1..n} > 0;
param d {1..n} > 0;

var X {1..n} >= 0, <= 2;
var Y {1..n} >= 0, <= 2;

maximize Obj:
(sum {j in 1..n} c[j]*X[j]) * (sum {j in 1..n} d[j]*Y[j]);

subject to SumX: sum {j in 1..n} j * X[j] >= 17;
subject to SumY: sum {j in 1..n} j * Y[j] >= 17;

subject to SumXY: sum {j in 1..n} (X[j] + Y[j]) = 7;

Product of Linear Terms

AMPL Solution

Rejected by Gurobi

ampl: model xy4.mod;
ampl: option solver gurobi;

ampl: solve;

Gurobi 5.5.0: quadratic objective is not positive definite

Product of Linear Terms

Solved by KNITRO

ampl: model xy4a.mod;
ampl: option solver knitro;

ampl: solve;

KNITRO 8.1.1: Locally optimal solution.
objective 887.414414; feasibility error 7.05e-08
10 iterations; 11 function evaluations

AMPL Model
Conic reformulation

param n > 0;
param c {1..n} > 0;
param d {1..n} > 0;

var X {1..n} >= 0, <= 2;
var Y {1..n} >= 0, <= 2;

var ZX >= 0;
var ZY >= 0;
var Z;

maximize Obj: Z;

subject to ZXdef: ZX = sum {j in 1..n} c[j]*X[j];
subject to ZYdef: ZY = sum {j in 1..n} d[j]*Y[j];

subject to Zdef: Z^2 <= ZX * ZY; # still not positive semidefinite

subject to SumX: .......

Product of Linear Terms

AMPL Solution

Now solved by Gurobi

ampl: model xy4b.mod;
ampl: option solver gurobi;

ampl: solve;

Gurobi 5.5.0: optimal solution; objective 29.78950153
11 barrier iterations

ampl: ampl: print Z^2;
887.4144013356272

Product of Linear Terms

Related cases
 Minimize can’t be reformulated

 (∑ ௝)௡௝ୀଵݔ ଵ/ଶ(∑ ௡௝ୀଵ(ݕ ଵ/ଶ offers more possibilities

 Many other products of powers can be handled

Givenܰ Set of nodes representing intersections݁ Entrance to network݂ Exit from networkܣ ⊆  ܰ ∪ {݁}  ×  ܰ ∪ {݂}
Set of arcs representing road links

andܾ݆݅ Base travel time for each road link (݅, ݆) ∈ ݆݅ݏܣ Traffic sensitivity for each road link (݅, ݆) ∈ ݆݅ܿܣ Capacity for each road link (݅, ݆) ∈ ܶܣ Desired throughput from ݁ to ݂
15

Example 2: Traffic Network

Determine݆݅ݔ Traffic flow through road link (݅, ݆) ∈ ݆݅ݐܣ Actual travel time on road link (݅, ݆) ∈ ܣ
to minimize

Average travel time from e to f

16

Formulation
Traffic Network

Subject to

17

Formulation (cont’d)

Travel times increase as flow approaches capacity

Flow out equals flow in at any intersection

Flow into the entrance equals the specified throughput

Traffic Network

AMPL Formulation

Symbolic data

set INTERS;          # intersections (network nodes)

param EN symbolic;   # entrance
param EX symbolic;   # exit

check {EN,EX} not within INTERS;

set ROADS within {INTERS union {EN}} cross {INTERS union {EX}};

# road links (network arcs)

param base {ROADS} > 0;  # base travel times
param sens {ROADS} > 0;  # traffic sensitivities
param cap {ROADS} > 0;   # capacities

param through > 0;       # throughput

Traffic Network

AMPL Formulation (cont’d)

Symbolic model

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Time {ROADS} >= 0;

minimize Avg_Time:
(sum {(i,j) in ROADS} Time[i,j] * Flow[i,j]) / through;

subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = base[i,j] + (sens[i,j]*Flow[i,j]) / (1-Flow[i,j]/cap[i,j]);

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Data

Explicit data independent of symbolic model

set INTERS := b c ;

param EN := a ;
param EX := d ;

param: ROADS: base cap sens :=
a b    4   10   .1
a c    1   12   .7
c b    2   20   .9
b d    1   15   .5
c d    6   10   .1 ;

param through := 20 ;

Traffic Network

a

c

b
d

AMPL Solution

Model + data = problem to solve, using KNITRO

ampl: model traffic.mod;
ampl: data traffic.dat;

ampl: option solver knitro;
ampl: solve;

KNITRO 8.1.1: Locally optimal solution.
objective 61.04695019; feasibility error 1.42e-14
9 iterations; 15 function evaluations

ampl: display Flow, Time;

:       Flow       Time   :=
a b    9.55146   25.2948
a c   10.4485    57.5709
b d   11.0044    21.6558
c b    1.45291    3.41006
c d    8.99562   14.9564
;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX?

ampl: model traffic.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.5.1.0: 
Constraint _scon[1] is not convex quadratic 
since it is an equality constraint.

Traffic Network



Robert Fourer, Jared Erickson, Convex Quadratic Programming in AMPL
ICCOPT 2013 — Lisbon 29 July-1 August 2013 24

AMPL Solution (cont’d)

Look at the model again . . .

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Time {ROADS} >= 0;

minimize Avg_Time:
(sum {(i,j) in ROADS} Time[i,j] * Flow[i,j]) / through;

subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = base[i,j] + (sens[i,j]*Flow[i,j]) / (1-Flow[i,j]/cap[i,j]);

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Quadratically constrained reformulation

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Delay {ROADS} >= 0;

minimize Avg_Time:
sum {(i,j) in ROADS} (base[i,j]*Flow[i,j] + Delay[i,j]) / through;

subject to Delay_Def {(i,j) in ROADS}:
sens[i,j] * Flow[i,j]^2 <= (1 - Flow[i,j]/cap[i,j]) * Delay[i,j];

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX?

ampl: model trafficQUAD.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.5.1.0: 
QP Hessian is not positive semi-definite.

Traffic Network

AMPL Solution (cont’d)

Quadratic reformulation #2

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Delay {ROADS} >= 0;
var Slack {ROADS} >= 0;

minimize Avg_Time:
sum {(i,j) in ROADS} (base[i,j]*Flow[i,j] + Delay[i,j]) / through;

subject to Delay_Def {(i,j) in ROADS}:
sens[i,j] * Flow[i,j]^2 <= Slack[i,j] * Delay[i,j];

subject to Slack_Def {(i,j) in ROADS}:
Slack[i,j] = 1 - Flow[i,j]/cap[i,j];

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX!

ampl: model trafficSOC.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.5.1.0: primal optimal; objective 61.04693968
15 barrier iterations

ampl: display Flow;

Flow :=
a b    9.55175
a c   10.4482
b d   11.0044
c b    1.45264
c d    8.99561
;

Traffic Network

Where they are done now
 In AMPL

 In the AMPL-solver interface

 In the solver

Where they should be done

How we have extended them

30

Detection and Transformation

Model instantiated with data

Expression trees written to problem file
 ହݕ5.3) (ଶݔ10.1) + (଼ݕ1.7
 (10.1 * x[2]) * (5.3 * y[5] + 1.7 * y[8])

31

In AMPL

*

+*

* *

5.3 y[5] 1.7 y[8]

10.1 x[2]

Quadratic problem detected
 Products of linear terms multiplied out

 Quadraticity test applied by recursive tree walk

Nonzero quadratic coefficients sent to solver
 Coefficients extracted from tree

 Solver-specific routines called

32

In the AMPL-Solver Interface

Test for recognized convex quadratics

“Elliptic” case: numerical test


Min ݔ்ܳݔ + ݔ்ܳݔݔݍ ≤ ݔݍ + ܿቋ where ܳ is positive semidefinite

“Conic” case: symbolic test
 . + ଵଶݔ . . ≥ ௡ଶݔ +  ௡ାଵଶݔ , ௡ାଵݔ ≥ 0
 . + ଵଶݔ . . ≥ ௡ଶݔ +  ௡ାଵݔ  ௡ାଵݔ ,௡ାଶݔ ≥ ௡ାଶݔ ,0 ≥ 0

. . . second-order cone programs (SOCPs)

33

In the Solver

In AMPL?
 Some solution strategies may be ruled out

In the solver?
 Each solver will have its own implementation

In the AMPL-solver interface?
 Recognition routines can be shared where appropriate

 Representation details can be different for each solver

 New ideas can be tried out

. . . interface source is open

34

Where Should Detection and
Transformation Be Done?



Robert Fourer, Jared Erickson, Convex Quadratic Programming in AMPL
ICCOPT 2013 — Lisbon 29 July-1 August 2013 35

Example 3: Schittkowski #255 (err)

var x {1..4} >= -20, <= 20;

minimize f: 

100*(x[2] - x[1]^2) + (1-x[1])^2 + 90*(x[4]-x[3]^2) + (1-x[3])^2 + 

10.1*((x[2]-1)^2 + (x[4]-1)^2) + 19.8*(x[2]-1)*(x[4]-1);

AMPL Solution by KNITRO

Starting point 2

ampl: model s255.mod;

ampl: let {j in 1..4} x[j] := +1;

ampl: solve;

KNITRO 8.0.0: Locally optimal solution.
objective -75376.125; feasibility error 0
8 iterations; 9 function evaluations

s255 (err)

Starting point 1

ampl: model s255.mod;

ampl: let {j in 1..4} x[j] := -1;

ampl: solve;

KNITRO 8.0.0: Locally optimal solution.
objective -75216.1247; feasibility error 0
7 iterations; 8 function evaluations

AMPL Solution by “Linear” Solvers

Solved by Gurobi

ampl: model s255.mod;

ampl: option solver gurobi;

ampl: solve;

Gurobi 5.5.0: optimal solution; objective -75376.125
7 barrier iterations

s255 (err)

Rejected by CPLEX

ampl: model s255.mod;

ampl: option solver cplex;

ampl: solve;

CPLEX 12.5.1.0: QP Hessian is not positive semi-definite.

Theory
 Targets for transformation

 SOCP-equivalent forms

Implementation via recursive tree walks
 Detection

 Transformation

Testing
 Existence of SOCP-equivalent problems

 Performance of “linear” vs. nonlinear solvers

Prospects . . .

38

Detection and Transformation 
of SOCP-Equivalent Forms

Standard cone

39

Theory: Conic Constraint Forms

. . . boundary not smooth

Rotated cone
 ଶݔ ≤ ,ݖݕ ݕ ≥ 0, ݖ ≥ 0, . . .

y

z

ଶݔ + ଶݕ ≤ ଶݖ ݖ ≥ 0 ଶݔ + ଶݕ ≤ ,ଶݖ ݖ ≥ 0

Symbolic detection
 . + ଵଶݔ . . ≥ ௡ଶݔ +  ௡ାଵଶݔ , ௡ାଵݔ ≥ 0
 . + ଵଶݔ . . ௡ାଵݔ ≥ ௡ଶݔ +  ௡ାଵݔ ,௡ାଶݔ ≥ ௡ାଶݔ ,0 ≥ 0

 implemented through recursive tree walks

Numerical detection
 ܠۿ்ܠ + ܠܙ ≤ where Q  ,ݎ has one negative eigenvalue

 see Ashutosh Mahajan and Todd Munson, “Exploiting 
Second-Order Cone Structure for Global Optimization”

 not addressed in our work

40

Targets for Transformation



Robert Fourer, Jared Erickson, Convex Quadratic Programming in AMPL
ICCOPT 2013 — Lisbon 29 July-1 August 2013

Quadratic
 Constraints

 Objectives

SOC-representable
 Quadratic-linear ratios

 Generalized geometric means

 Generalized p-norms

Other objective functions
 Generalized product-of-powers

 Logarithmic Chebychev

41

SOCP-Equivalent Forms

Standard cone constraints
 ∑ ܽ௜(܎௜ܠ + ݃௜)ଶ௡௜ୀଵ ≤ ܽ௡ାଵ(܎௡ାଵܠ + ݃௡ାଵ)ଶ,ܽଵ, . . . , ܽ௡ାଵ ≥ 0, ܠ௡ାଵ܎ + ݃௡ାଵ ≥ 0

 ∑ ௜ଶݒ ≤ ௡ାଵ,௡௜ୀଵݒ ௡ାଵݒ ≥ 0
 ௜ݒ = ܽ௜ଵ/ଶ(܎௜ܠ + ݃௜),  ݅ = 1, . . . , ݊ + 1

Rotated cone constraints
 ∑ ܽ௜ ܠ௜܎ + ݃௜ ଶ௡௜ୀଵ ≤ ܽ௡ାଵ ܠ௡ାଵ܎ + ݃௡ାଵ ܠ௡ାଶ܎ + ݃௡ାଶ ,ܽଵ, . . . , ܽ௡ାଵ ≥ 0, ܠ௡ାଵ܎   + ݃௡ାଵ ≥ ܠ௡ାଶ܎  ,0 + ݃௡ାଶ ≥ 0

Sum-of-squares objectives
 Minimize  ∑ ܽ௜ ܠ௜܎ + ݃௜ ଶ௡௜ୀଵ

 Minimize ݒ
Subject to ∑ ܽ௜ ܠ௜܎ + ݃௜ ଶ௡௜ୀଵ ≤ ݒ  ,ଶݒ ≥ 0

Quadratic Generalizations
SOCP-equivalent

Definition
 Function ݏ ݔ is SOC-representable iff . . .

 ݏ ݔ ≤ ܽ௡ାଵ(܎௡ାଵܠ + ݃௡ାଵ) is equivalent to some 
combination of linear and quadratic cone constraints

Minimization property
 Minimize ݏ ݔ is SOC-equivalent

 Minimize ௡ାଵݒ
Subject to (ݔ)ݏ ≤ ௡ାଵݒ

Combination properties
 ܽ ∙ ݏ ݔ is SOC-representable for any ܽ ≥ 0
 ∑ ௜ݏ ௡௜ୀଵݔ is SOC-representable

 ௜ୀଵ௡ݔܽ݉ ௜ݏ  ݔ is SOC-representable
. . . requires a recursive detection algorithm!

43

SOC-Representable
SOCP-equivalent

SOC-Representable (1)
Vector norm

 ܉ ∙ ܠ۴ + ܏ = ∑ ܽ௜ଶ ܠ௜܎ + ݃௜ ଶ௡௜ୀଵ ≤ ܽ௡ାଵ ܠ௡ାଵ܎ + ݃௡ାଵ
 square both sides to get standard SOC∑ ܽ௜ଶ ܠ௜܎ + ݃௜ ଶ௡௜ୀଵ ≤ ܽ௡ାଵଶ ܠ௡ାଵ܎ + ݃௡ାଵ ଶ

Quadratic-linear ratio


∑ ௔೔ ା ௚೔ ܠ೔܎ మ೙೔సభ܎೙శమܠ ା ௚೙శమ ≤ ܽ௡ାଵ ܠ௡ାଵ܎ + ݃௡ାଵ
 where ܎௡ାଶܠ + ݃௡ାଶ ≥ 0
 multiply by denominator to get rotated SOC∑ ܽ௜ ܎௜ܠ + ݃௜ ଶ௡௜ୀଵ ≤ ܽ௡ାଵ ܎௡ାଵܠ + ݃௡ାଵ ܠ௡ାଶ܎ + ݃௡ାଶ

SOC-Representable (2)
Negative geometric mean

 − ∏ ܠ௜܎ + ݃௜ ଵ ௣⁄௣௜ୀଵ ≤ ܠ௡ାଵ܎ + ݃௡ାଵ,  ݌ ∈ ℤା
 ସଵ/ସݔଷଵ/ସݔଶଵ/ସݔଵଵ/ସݔ− ≤ ହݔ− becomes rotated SOCs:ݔହଶ ≤ ଵଶݒ ,ଶݒଵݒ ≤ ଶଶݒ ,ଶݔଵݔ ≤ ସݔଷݔ
 apply recursively logଶ ݌ times

Generalizations
 − ∏ ܠ௜܎ + ݃௜ ఈ೔௡௜ୀଵ ≤ ܽ௡ାଵ ܠ௡ାଵ܎ + ݃௡ାଵ :  ∑ ௜ߙ ≤ 1௡௜ୀଵ ௜ߙ , ∈ ℚା
 ∏ ܠ௜܎ + ݃௜ ିఈ೔௡௜ୀଵ ≤ ܽ௡ାଵ ܠ௡ାଵ܎ + ݃௡ାଵ ௜ߙ  , ∈ ℚା

 all require ܎௜ܠ + ݃௜ to have proper sign

SOC-Representable (3)
p-norm

 ∑ ܠ௜܎ + ݃௜ ௣௡௜ୀଵ ଵ ௣⁄ ≤ ܠ௡ାଵ܎ + ݃௡ାଵ,  ݌ ∈ ℚା,  ݌ ≥ 1
 ଵݔ ହ + ଶݔ ହ ଵ ହ⁄ ≤ ଷݔ can be writtenݔଵ ହ ⁄ଷସݔ + ଶݔ ହ ⁄ଷସݔ  ≤ ଷݔ which becomesݒଵ + ଶݒ ≤ ଷݔ with  −ݒଵଵ ହ⁄ ଷସݔ ହ⁄ ≤ ଵଵݒ− ,ଵݔ± ହ⁄ ଷସݔ ହ⁄ ≤ ଶݔ±
 reduces to product of powers

Generalizations
 ∑ ܠ௜܎ + ݃௜ ఈ೔௡௜ୀଵ ଵ ఈబ⁄ ≤ ܠ௡ାଵ܎ + ݃௡ାଵ,  α௜ ∈ ℚା,  ߙ௜ ≥ ଴ߙ  ≥ 1
 ∑ ܠ௜܎ + ݃௜ ఈ೔௡௜ୀଵ ≤ ܠ௡ାଵ܎ + ݃௡ାଵ ఈబ
 Minimize  ∑ ܠ௜܎ + ݃௜ ఈ೔௡௜ୀଵ

. . . standard SOCP has ࢏ࢻ ≡ ૛

Other Objective Functions
Unrestricted product of powers

 Minimize  − ∏ ܠ௜܎ + ݃௜ ఈ೔௡௜ୀଵ for any ߙ௜ ∈ ℚା
Logarithmic Chebychev approximation

 Minimize  max௜ୀଵ௡ log ܠ௜܎ − log (݃௜)  
Why no constraint versions?

 Not SOC-representable

 Transformation changes objective value (but not solution)

Principles
 Representation of expressions by trees

 Recursive tree-walk functions
 isLinear(), isQuadratic(), buildLinear()

Example: Sum of norms

Expression
base[i,j] + (sens[i,j]*Flow[i,j]) / (1-Flow[i,j]/cap[i,j])

Expression tree

49

Representation

+

/5

* -

0.1 x[5] 1 /

x[5] 10

. . . actually a DAG

Detection: isQuadr()

boolean isQuadr (Node);

case of Node {

PLUS:  
MINUS: return( isQuadr(Node.left) and isQuadr(Node.right) );

TIMES: return( isLinear(Node.left) and isLinear(Node.right) or
isQuadr(Node.left) and isConst(Node.right)   or
isConst(Node.left) and isQuadr(Node.right) );

POWER: return( isLinear(Node.left) and 
isConst(Node.right) and value(Node.right) == 2 );

VAR: return( TRUE );

CONST: return( TRUE );

}

Principles

Detection: isLinear()

boolean isLinear (Node);

case of Node {

PLUS:  
MINUS: return( isLinear(Node.left) and isLinear(Node.right) );

TIMES: return( isConst(Node.left) and isLinear(Node.right) or
isLinear(Node.left) and isConst(Node.right) );

DIV: return( isLinear(Node.left) and isConst(Node.right) );

VAR: return( TRUE );

CONST: return( TRUE );

}

. . . to detect, test isLinear(root)

Principles

Transformation: buildLinear()

(coeff,const) = buildLinear (Node);

if Node.L then (coefL,consL) = buildLinear(Node.L);
if Node.R then (coefR,consR) = buildLinear(Node.R);

case of Node {

PLUS:  coeff = mergeLists( coefL, coefR );
const = consL + consR;

TIMES: ...

DIV: coeff = coefL / consR;
const = consL / consR;

VAR: coeff = makeList( 1, Node.index );
const = 0;

CONST: coeff = makeList( );
const = Node.value;

}

. . . to transform, call buildLinear(root)

Principles

Example: Sum-of-Norms Objective
Given

 Minimize  ∑ ܽ௜ ∑ ܠ௜௝܎ + ݃௜௝ ଶ௡೔௝ୀଵ௠௜ୀଵ
Transform to

 Minimize  ∑ ܽ௜ݕ௜௠௜ୀଵ
 ∑ ௜௝ଶ௡೔௝ୀଵݖ ≤ ௜ݕ  ,௜ଶݕ ≥ 0, ݅ = 1, . . . , ݉
 ௜௝ݖ = ܠ௜௝܎ + ݃௜௝, ݅ = 1, . . . , ݉, ݆ = 1, . . . , ݊௜

Two steps
 Detection

 Transformation

Detection

SUMOFNORMS

Sum: ݁ଵ + ݁ଶ is SUMOFNORMS if ݁ଵ, ݁ଶ are SUMOFNORMS

Product: ݁ଵ݁ଶ is SUMOFNORMS if݁ଵ is SUMOFNORMS and ݁ଶ is POSCONSTANT or݁ଶ is SUMOFNORMS and ݁ଵ is POSCONSTANT

Square root: ݁ is SUMOFNORMS if ݁ is SUMOFSQUARES

Sum of Norms

SUMOFSQUARES

Sum: ݁ଵ + ݁ଶ is SUMOFSQUARES if ݁ଵ, ݁ଶ are SUMOFSQUARES

Product: ݁ଵ݁ଶ is SUMOFSQUARES if݁ଵ is SUMOFSQUARES and ݁ଶ is POSCONSTANT or݁ଶ is SUMOFSQUARES and ݁ଵ is POSCONSTANT

Square: ݁ଶ is SUMOFSQUARES if ݁ is LINEAR

Constant: ܿ is SUMOFSQUARES if ܿ is POSCONSTANT

Mathematical

 Minimize  ∑ ܽ௜ ∑ ܠ௜௝܎ + ݃௜௝ ଶ௡೔௝ୀଵ௠௜ୀଵ
Practical

 Constant multiples inside any sum

 Recursive nesting of constant multiples & sums

 Constant as a special case of a square

 3 ଵݔ4 + 7 ଶݔ + ଷݔ2 + 6 ଶ + ସݔ + ହݔ ଶ + 17

55

Detection Issues
Sum of Norms



Robert Fourer, Jared Erickson, Convex Quadratic Programming in AMPL
ICCOPT 2013 — Lisbon 29 July-1 August 2013 56

Transformation

TRANSFORMSUMOFNORMS (Expr e, Obj o, real k)
Sum: e1 + e2 where e1, e2 are SUMOFNORMS

TRANSFORMSUMOFNORMS (e1,o,k)
TRANSFORMSUMOFNORMS (e2,o,k)

Product: e1 * c2 where e1 is SUMOFNORMS and c2 is POSCONSTANT

TRANSFORMSUMOFNORMS (e1,o,c2∗k)

Product: c1 * e2 where e2 is SUMOFNORMS and c1 is POSCONSTANT

TRANSFORMSUMOFNORMS (e2,o,c1∗k)

Square root: sqrt(e) where e is SUMOFSQUARES

yi := NEWNONNEGVAR();  o += k * yi
qi := NEWLECON();  qi += -yi^2
TRANSFORMSUMOFSQUARES (e,qi,1)

Sum of Norms

Transformation (cont’d)

TRANSFORMSUMOFSQUARES (Expr e, LeCon qi, real k)
Sum: e1 + e2 where e1, e2 are SUMOFSQUARES

TRANSFORMSUMOFSQUARES (e1,o,k)
TRANSFORMSUMOFSQUARES (e2,o,k)

Product: e1 * c2 where e1 is SUMOFSQUARES and c2 is POSCONSTANT

TRANSFORMSUMOFSQUARES (e1,o,c2∗k)

Product: c1 * e2 where e2 is SUMOFSQUARES and c1 is POSCONSTANT

TRANSFORMSUMOFSQUARES (e2,o,c1∗k)

Square: sqr(zij) where zij is VARIABLE

qi += k * zij^2
Square: sqr(e) where e is LINEAR

zij := NEWVAR();  qi += k * zij^2
lij := NEWEQCON();  lij += zij - e

Constant: c is POSCONSTANT

zij := NEWVAR();  qi += k * zij^2
lij := NEWEQCON();  lij += zij – sqrt(c)

Sum of Norms

Mathematical
 Minimize  ∑ ܽ௜ݕ௜௠௜ୀଵ
 ∑ ௜௝ଶ௡೔௝ୀଵݖ ≤ ௜ݕ  ,௜ଶݕ ≥ 0
 ௜௝ݖ = ܠ௜௝܎ + ݃௜௝

Practical
 Generalization: handle all previously mentioned

 Efficiency: don’t define ݖ௜௝ when ܎௜௝ܠ + ݃௜௝ is a single variable

 Trigger by calling TRANSFORMSUMOFNORMS(e,o,k) with
 e the root node
 o an empty objective
 k = 1

Challenges
Extending to all cases previously cited

 All prove amenable to recursive tree-walk

 Details much harder to work out

Checking nonnegativity of linear expressions
 Heuristic catches many non-obvious instances

Assessing usefulness . . .

Survey of nonlinear test problems

Comparison of performance

Survey of Test Problems (1)
12% of 1238 nonlinear problems were SOC-solvable!

 not counting QPs with sum-of-squares objectives

 from Vanderbei’s CUTE & non-CUTE, and netlib/ampl

A variety of forms detected
 hs064 has 4 ⁄ଵݔ + 32 ⁄ଶݔ + 120 ⁄ଷݔ ≤  1
 hs036 minimizes −ݔଵݔଶݔଷ
 hs073 has 1.645 0.28ݔଵଶ + ଶଶݔ0.19  + ଷଶݔ20.5  + ସଶݔ0.62  ≤ . . .
 polak4 is a max of sums of squares

 hs049 minimizes ݔଵ − ଶݔ ଶ + ଷݔ  − 1 ଶ + ସݔ  − 1 ସ + ହݔ  − 1 ଺
 emfl_nonconvex has ∑ ௝௞ݔ − ܽ௜௞ ଶଶ௞ୀଵ ≤ ௜௝ଶݏ 

Survey of Test Problems (2)
Counted number of test problems . . .

 Solvable already by a “linear” solver

 Detected as SOCP-equivalent by our routines

62

Comparison of Performance
SOCP-equivalent with nonsmooth functions

var x {1..5} integer;
var y {1..5} >= 0;

minimize obj: sum {i in 1..5} (

sqrt( (x[i]+2)^2 + (y[i]+1)^2 ) + sqrt( (x[i]+y[i])^2 ) + y[3]^2 );

subj to xsum: sum {i in 1..5} x[i] <= -12; 

subj to ysum: sum {i in 1..5} y[i] >= 10; 

subj to socprep: 

max {i in 1..5} ( (x[i]^2 + 1)/(i+y[i]) + y[i]^3 ) <= 30;

Comparison (cont’d)

General nonlinear solver (integer)

KNITRO 8.0.0: Convergence to an infeasible point. 

Problem may be locally infeasible.

General nonlinear solver (continuous relaxation)

KNITRO 8.0.0: 

--- ERROR evaluating objective gradient.
--- ERROR evaluating constraint gradients.

Evaluation error.

objective 17.14615551; feasibility error 0
233 iterations; 1325 function evaluations



Robert Fourer, Jared Erickson, Convex Quadratic Programming in AMPL
ICCOPT 2013 — Lisbon 29 July-1 August 2013 65

Comparison
Convex quadratic solver (integer)

CPLEX 12.4.0

Total time (root+branch&cut) = 0.21 sec.

Solution value  = 17.246212

:    x         y
1   -3   3
2   -2   1.99993
3   -2   0.000300084
4   -3   3
5   -2   1.99993
;

Comparison
Convex quadratic solver (continuous relaxation)

CPLEX 12.4.0

Total time = 0.04 sec.

Solution value  = 17.141355

:   x         y
1   -2.49707   2.49707
2   -2.49707   2.49707
3   -2.01171   0.011716
4   -2.49707   2.49707
5   -2.49707   2.49707
;

Robust implementation needed
 Focus on forms most likely to be worthwhile

 Modularize to work with varied solver interfaces

Value will be established gradually
 Teach users about convex quadratic features

 Collect experience with new models

67

